

**FOR****PURPOSE**

Specify row based subsets or intervals for functions.

**DESCRIPTION**

The FOR keyword serves 4 separate purposes:

1. At the end of the PLOT commands (when plotting a function), it specifies the points of the function that are to be computed and plotted. For example, PLOT SIN(X) FOR X = 0 .1 6 plots the sine function starting with  $x = 0$ , at increments of .1, and stopping at the value 6. That is, at the points, 0, .1, .2, .3, ..., 5.9, 6.0.
2. At the end of the 3D-PLOT command (when plotting a surface function), it specifies the points of the function that are to be computed and plotted. For example, PLOT EXP(-X\*X-Y\*Y) FOR X = -2 .2 2 FOR Y = -2 .2 2 plots the bivariate normal function with cross-hatch lines at  $x = -2.0, -1.8, -1.6, \dots, 1.8, 2.0$ , and at  $y = -2.0, -1.8, -1.6, \dots, 1.8, 2.0$ .
3. At the end of all graphical and analysis commands, and certain support commands, it specifies the elements (= "rows") of a variable to be operated on. For example, PRINT X FOR I = 1 10 1000 prints the variables X starting with element 1, at increments of 10 elements, and stopping at element 1000. That is, it prints X(1), X(11), X(21), ..., X(981), X(991).
4. At the end of the LET sub-commands ROOTS, DERIVATIVE, and INTEGRAL, it specifies limits or evaluation points. For example, LET R = ROOTS EXP(-X)+SIN(X\*\*2) FOR X = 0 TO 5 finds all roots of the function in the domain  $X = 0$  to  $X = 5$ .

**SYNTAX 1**

PLOT <f> FOR <x> = <x1> <xinc> <x2>

where <f> is an explicit functional expression or a function name;

<x> is the name of the dummy variable in the function (the horizontal axis variable);

<x1> is a number or parameter that is the minimum value for the dummy variable;

<xinc> is a number or parameter that is the increment for the dummy variable;

and <x2> is a number or parameter that is the maximum value for the dummy variable.

This syntax is used for plotting functions with the PLOT command. For example,

```
PLOT SIN(X) FOR X = 0 0.1 10
```

**SYNTAX 2**

3D-PLOT <f> FOR <x> = <x1> <inc> <x2> FOR <y> = <y1> <inc> <y2>

where <f> is an explicit functional expression or a function name;

<x> is the name of one dummy variable in the function (one horizontal axis variable);

<x1> is a number or parameter that is the minimum value for this dummy variable;

<xinc> is a number or parameter that is the increment for this dummy variable;

<x2> is a number or parameter that is the maximum value for this dummy variable;

<y> is the name of the other dummy variable in the function (the other horizontal axis variable);

<y1> is a number or parameter that is the minimum value for this dummy variable;

<yinc> is a number or parameter that is the increment for this dummy variable;

and <y2> is a number or parameter that is the maximum value for this dummy variable.

This syntax is used for plotting 3-d functions with the 3D-PLOT command. For example,

```
PLOT EXP(-X*X - Y*Y) FOR X = -2 0.2 2 FOR Y = -2 0.2 2
```

**SYNTAX 3**

<Command> FOR I = <i1> <inc> <i2>

where <Command> is any graphics or analysis command or certain support commands;

<i1> is a number or parameter that is the element/row designation of the first element to be operated on;

<inc> is a number or parameter that is the increment for the element/row;

and <i2> is a number or parameter that is the element/row designation of the last element to be operated on.

This syntax can be used with all graphics and analysis commands and with many support commands. With this syntax, the dummy index variable must be I (as opposed to FOR J or some other variable) or unpredictable results may occur. An example of this syntax is

```
PRINT X FOR I = 1 10 1000
```

## SYNTAX 4

```
LET <y> = ROOTS <f> FOR <x> = <x1> TO <x2>
LET <y> = DERIVATIVE <f> WRT <x> FOR <x> = <x1>
LET <y> = INTEGRAL <f> WRT <x> FOR <x> = <x1> TO <x2>
```

where <y> is a parameter or variable name;

<f> is an explicit functional expression or a function name;

<x> is the name of the dummy variable in the function (the horizontal axis variable);

<x1> is a number or parameter that is

- a) (for ROOTS) the minimum point in the search region;
- b) (for DERIVATIVE) the point at which the derivative is to be evaluated;
- c) (for INTEGRAL) the lower limit of integration;

and <x2> is a number or parameter that is

- a) (for ROOTS) the maximum point in the search region;
- b) (for INTEGRAL) the upper limit of integration.

This syntax is used with certain LET sub-commands. For example,

```
LET R = ROOTS EXP(-X) + SIN(X**2) FOR X = 0 TO 5
```

## EXAMPLES

```
PLOT SIN(X) FOR X = 0 .1 6
PLOT F FOR X = A DEL B
3D-PLOT X+Y FOR X = 0 1 5 FOR Y = 0 1 5
3D-PLOT G FOR X = A1 DEL1 B1 FOR Y = A2 DEL2 B2
HISTOGRAM Y FOR I = 20 1 50
FIT (A+B*X)/(C+D*X) FOR I = 101 1 200
SMOOTH Y FOR I = 200 1 500
RETAIN Y FOR I = 10 10 200
LET R=ROOTS SIN(X**2)*EXP(-X) WRT X FOR X = 0 TO 6
LET A = DERIVATIVE F1 WRT X FOR X = 1.3
LET A = INTEGRAL LOG(X) WRT X FOR X = 1 TO 2
```

## DEFAULT

None

## SYNONYMS

None

## RELATED COMMANDS

PLOT	=	Generates a data or function plot.
3D-PLOT	=	Generates a 3-d data or function plot.
LET	=	Carries out math operations (and many other operations).
SUBSET	=	Allows specification of a subset.
EXCEPT	=	Allows exclusion-specification of a subset.

## APPLICATIONS

Data subsets

## IMPLEMENTATION DATE

Pre-1987

## PROGRAM

```
LET R = ROOTS EXP(-X) + SIN(X**2) FOR X = 0 TO 5
PRINT R
```